
Show that Bus who's Boss!
ModBoss is an Elixir library that maps Modbus registers to human-friendly names and provides
automatic encoding/decoding of values—making your application logic simpler and more readable,
and making testing of modbus concerns easier.
Note that ModBoss doesn't handle the actual reading/writing of modbus registers—it simply assists
in providing friendlier access to register values. You'll likely be wrapping another library such
as Modbux for the actual reads/writes.
Installation
If available in Hex, the package can be installed
by adding modboss to your list of dependencies in mix.exs:
def deps do
  [
    {:modboss, "~> 0.1.0"}
  ]
end
Usage
1. Map your schema
Starting with the type of register, you'll define the addresses to include and a friendly name
for the mapping.
The :as option dictates how values will be encoded before being written to Modbus or decoded
after being read from Modbus. You can use translation functions from another module—like those
found in ModBoss.Encoding—or provide your own as shown here with as: :fw_version.
When providing your own translation functions, ModBoss expects that you'll provide functions
corresponding to the :as option but with encode_ / decode_ prefixes added as applicable.
These functions will receive the value to be translated and should return either
{:ok, translated_value} or {:error, message}.
defmodule MyDevice.Schema do
  use ModBoss.Schema

  modbus_schema do
    holding_register 1, :outdoor_temp, as: {ModBoss.Encoding, :signed_int}
    holding_register 2..5, :model_name, as: {ModBoss.Encoding, :ascii}
    holding_register 6, :version, as: :fw_version, mode: :rw
    # Also supports: input_register / coil / discrete_input
  end

  def encode_fw_version(value) do
    encoded_value = do_encode(value)
    {:ok, encoded_value}
  end

  def decode_fw_version(value) do
    decoded_value = some_decode_logic(value)
    {:ok, decoded_value}
  end
end
In this example:
	Holding register at address 1 is named outdoor_temp and uses the built-in signed_int
decoder that ships with ModBoss.
	Holding registers 2–5 are grouped under the name model_name and use a built-in ASCII
decoder.
	Holding register 6 is named version and uses encode_fw_version/1 and decode_fw_version/1
to translate values being written or read respectively.

2. Provide generic read/write functions
You'll need to provide a read_func/3 and a write func/3 for actually
interacting on the Modbus. In practice, these functions will likely build on a library like
Modbux along with state stored in a GenServer (e.g.
a modbux_pid, IP Address, etc.) to perform the read/write operations.
For each batch, the read_func will be provided the type of register
(:holding_register, :input_register, :coil, or :discrete_input), the starting address,
and the number of addresses to read. It must return either {:ok, result} or {:error, message}.
read_func = fn register_type, starting_address, count ->
  result = custom_read_logic(…)
  {:ok, result}
end
For each batch, the write_func will be provided the type of register (:holding_register or
:coil), the starting address for the batch to be written, and a list of values to write.
It must return either :ok or {:error, message}.
write_func = fn register_type, starting_address, value_or_values ->
  result = custom_write_logic(…)
  {:ok, result}
end
3. Read & Write by name!
From here you can read and write by name…
Requesting a single value returns just one value:
iex> ModBoss.read(MyDevice.Schema, read_func, :outdoor_temp)
{:ok, 72}
Requesting multiple values returns a map:
iex> ModBoss.read(MyDevice.Schema, read_func, [:outdoor_temp, :model_name, :version])
{:ok, %{outdoor_temp: 72, model_name: "AI4000", version: "0.1"}}
Writing is performed via a keyword list or map:
iex> ModBoss.write(MyDevice.Schema, write_func, version: "0.2")
:ok
Benefits
Extracting your Modbus schema allows you to isolate the encode/decode logic
making it much more testable. Your primary application logic becomes simpler and more
readable since it references registers by name and doesn't need to worry about encoding/decoding
of values. It also becomes fairly straightforward to set up virtual devices with the exact
same register mappings as your physical devices (e.g. using an Elixir Agent to hold the state of
the registers in a map). And it makes for easier troubleshooting since you don't need to
memorize (or look up) the register mappings when you're at an iex prompt.

          @type read_func() :: (register_type(),
                starting_address :: integer(),
                count :: integer() ->
                  {:ok, any()} | {:error, any()})

                 starting_address :: integer(),

                   :ok | {:error, any()})

          @spec read(module(), read_func(), atom() | [atom()], keyword()) ::
  {:ok, any()} | {:error, any()}


      


Read from modbus using named mappings.
This function takes either an atom or a list of atoms representing the mappings to read,
batches the mappings into contiguous addresses per type, then reads and decodes the values
before returning them.
For each batch, read_func will be called with the type of register (:holding_register,
:input_register, :coil, or :discrete_input), the starting address for the batch
to be read, and the count of addresses to read from. It must return either {:ok, result}
or {:error, message}.
If a single name is requested, the result will be an :ok tuple including the singule result
for that named mapping. If a list of names is requested, the result will be an :ok tuple
including a map with mapping names as keys and mapping values as results.
Opts
	:decode — if false, returns the "raw" result as provided by read_func; defaults to true

Examples
read_func = fn register_type, starting_address, count ->
  result = custom_read_logic(…)
  {:ok, result}
end

# Read one mapping
ModBoss.read(SchemaModule, read_func, :foo)
{:ok, 75}

# Read multiple mappings
ModBoss.read(SchemaModule, read_func, [:foo, :bar, :baz])
{:ok, %{foo: 75, bar: "ABC", baz: true}}

# Read *all* readable mappings
ModBoss.read(SchemaModule, read_func, :all)
{:ok, %{foo: 75, bar: "ABC", baz: true, qux: 1024}}

# Get "raw" Modbus values (as returned by `read_func`)
ModBoss.read(SchemaModule, read_func, :all, decode: false)
{:ok, %{foo: 75, bar: [16706, 17152], baz: 1, qux: 1024}}

          @spec write(module(), write_func(), values_to_write()) :: :ok | {:error, any()}


      


Write to modbus using named mappings.
ModBoss automatically encodes your values, then batches any encoded values destined for
contiguous registers—creating separate batches per register type.
For each batch, write_func will be called with the type of register (:holding_register or
:coil), the starting address for the batch to be written, and a list of values to write.
It must return either :ok or {:error, message}.
Batch values
Each batch will contain either a list or an individual value based on the number of
addresses to be written—so you should be prepared for both.
Non-atomic writes!
While ModBoss.write/3 has the feel of being atomic, it's important to recognize that it
is not! It's fully possible that a write might fail after prior writes within the same call to
ModBoss.write/3 have already succeeded.
Within ModBoss.write/3, if any call to write_func returns an error tuple,
the function will immediately abort, and any subsequent writes will be skipped.
Example
write_func = fn register_type, starting_address, value_or_values ->
  result = custom_write_logic(…)
  {:ok, result}
end

iex> ModBoss.write(MyDevice.Schema, write_func, foo: 75, bar: "ABC")
:ok

Built-in encoding/decoding functions to get you started.
To make use of these functions, use the :as option in your ModBoss.Schema but leave off
the encode_ or decode_ prefix.
In other words, to use the built-in ASCII translation, specifiy as: :ascii in your schema.
Note about that extra arg…
Note that for built-in encode_* functions, we pass not just the value but also the mapping
itself. That's why you'll see an extra argument passed to these encoders!
We do this in order to provide more generic and helpful functions out the box (like
encoding of ASCII, which requires knowledge of how many registers we're encoding for). However,
when providing your own encode_* functions, you'll only be passed the value to be encoded
(and not the mapping).

        Decode value to an unsigned integer.

          @spec decode_ascii(integer() | [integer()]) :: {:ok, binary()}


      


Decode integer value(s) representing ASCII characters to text
	Up to 2 ASCII characters are stored per register.
	0 is interpreted as a terminator.

Examples
iex> decode_ascii(18537)
{:ok, "Hi"}

iex> decode_ascii([18537, 8448, 0])
{:ok, "Hi!"}

          @spec decode_boolean(integer()) :: {:ok, boolean()} | {:error, binary()}


      


Interpret 1 as true and 0 as false

  



  
    
      
    
    
      decode_signed_int(value)



    

  


  

      

          @spec decode_signed_int(integer()) :: {:ok, integer()}


      


Decode value to a signed integer.
Examples
iex> decode_signed_int(77)
{:ok, 77}

iex> decode_signed_int(65_535)
{:ok, -1}

          @spec decode_unsigned_int(integer()) :: {:ok, integer()}


      


Decode value to an unsigned integer.
Examples
iex> decode_unsigned_int(65_535)
{:ok, 65_535}

          @spec encode_ascii(binary(), ModBoss.Mapping.t()) ::
  {:ok, [integer()]} | {:error, binary()}


      


Encode text to integers representing ASCII characters.
	Prepares zero or more characters to be stored in contiguous registers.
	Up to 2 ASCII characters can be stored per register.
	0 is used as a terminator if fewer than the maximum characters are encoded.
	The ModBoss.Mapping is required in order to determine how many characters are supported.

Examples
iex> encode_ascii("Hi!", %ModBoss.Mapping{register_count: 3})
{:ok, [18537, 8448, 0]}

iex> {:error, _too_many_characters} = encode_ascii("Hi!", %ModBoss.Mapping{register_count: 1})

          @spec encode_boolean(boolean(), ModBoss.Mapping.t()) ::
  {:ok, integer()} | {:error, binary()}


      


Encode true as 1 and false as 0

          @spec encode_signed_int(integer(), ModBoss.Mapping.t()) ::
  {:ok, integer()} | {:error, binary()}


      


Encode value as a signed integer.
Valid values are -32,768 to 32,767.
This function assumes the expected output is a regular Elixir integer. It simply provides a
guard against overly large values, then returns the provided value.
Examples
iex> {:ok, -32_768} = encode_signed_int(-32_768, %{})
iex> {:error, _too_large} = encode_signed_int(32_768, %{})

          @spec encode_unsigned_int(integer(), ModBoss.Mapping.t()) ::
  {:ok, integer()} | {:error, binary()}


      


Encode value as an unsigned integer.
Valid values are 0 to 65,535.
This function assumes the expected output is a regular Elixir integer. It simply provides a
guard against overly large values, then returns the provided value.
Examples
iex> {:ok, 65_535} = encode_unsigned_int(65_535, %{})
iex> {:error, _too_large} = encode_unsigned_int(65_536, %{})

Struct representing the Modbus mapping.

          @type t() :: %ModBoss.Mapping{
  addresses: Range.t(),
  as: atom() | {module(), atom()},
  encoded_value: integer(),
  mode: :r | :rw | :w,
  name: atom(),
  register_count: integer(),
  starting_address: integer(),
  type: :holding_register | :input_register | :coil | :discrete_input,
  value: any()
}

Macros for establishing Modbus schema.
The schema allows names to be assigned to individual registers or groups of contiguous
registers along with encoder/decoder functions. It also allows registers to be flagged
as readable and/or writable.
Naming an address
You'll name a Modbus address with this format:
holding_register 17, :outdoor_temp, as: {ModBoss.Encoding, :signed_int}
This establishes address 17 as a holding register with the name :outdoor_temp.
The raw value from the register will be passed to ModBoss.Encoding.decode_signed_int/1
before being returned.
Similarly, to set aside a group of registers to hold a single logical value,
it would look like:
holding_register 20..23, :model_name, as: {ModBoss.Encoding, :ascii}
This establishes addresses 20–23 as holding registers with the name :model_name. The raw
values from these registers will be passed (as a list) to ModBoss.Encoding.decode_ascii/1
before being returned.
Mode
All registers are read-only by default. Use mode: :rw to allow both reads & writes.
Or use mode: :w to mark a register as write-only.
Automatic encoding/decoding
Depending on whether a mapping is flagged as readable/writable, it is expected that you
will provide functions with encode_ or decode_ prepended to the value provided by the :as
option.
For example, if you specify as: :on_off for a read-only register, ModBoss will expect that
the schema module defines an encode_on_off/1 function that accepts the value to encode and
returns either {:ok, encoded_value} or {:error, messsage}.
If the function to be used lives outside of the current module, a tuple including the module
name can be passed. For example, you can use built-in translation from ModBoss.Encoding such
as :boolean, :signed_int, :unsigned_int, and :ascii.
output of encode_*
Your encode function may need to encode for one or multiple registers, depending on the
mapping. You are free to return either a single value or a list of values—the important thing
is that the number of values returned needs to match the number of registers for your mapping.
If it doesn't, ModBoss will detect that and return an error during encoding.
For example, if encoding "ABC!" as ascii into a mapping with 3 registers, this would
technically only "require" 2 registers (one 16-bit register for every 2 characters).
However, your encoding should return a list of 3 values if that's what you've assigned
to the mapping in your schema.
input to decode_*
When decoding a single register, the decode function will be passed the single value from that
register as provided by your read function.
When decoding multiple registers (e.g. in ModBoss.Encoding.decode_ascii/1), the decode
function will be passed a List of values.
Example
defmodule MyDevice.Schema do
  use ModBoss.Schema

  modbus_schema do
    holding_register 1..5, :model, as: {ModBoss.Encoding, :ascii}
    holding_register 6, :outdoor_temp, as: {ModBoss.Encoding, :signed_int}
    holding_register 7, :indoor_temp, as: {ModBoss.Encoding, :unsigned_int}

    input_register 200, :foo, as: {ModBoss.Encoding, :unsigned_int}
    coil 300, :bar, as: :on_off, mode: :rw
    discrete_input 400, :baz, as: {ModBoss.Encoding, :boolean}
  end

  def encode_on_off(:on), do: {:ok, 1}
  def encode_on_off(:off), do: {:ok, 0}

  def decode_on_off(1), do: {:ok, :on}
  def decode_on_off(0), do: {:ok, :off}
end

Adds a coil to a schema.
Opts
	:mode — Makes the mapping readable/writable — can be one of [:r, :rw, :w] (default: :r)
	:as — Determines which encoding/decoding functions to use when writing/reading values.
See explanation of automatic encoding/decoding.

Adds a read-only discrete input to a schema.
Opts
	:as — Determines which decoding functions to use when reading values.
See explanation of automatic encoding/decoding.

Adds a holding register to a schema.
Opts
	:mode — Makes the mapping readable/writable — can be one of [:r, :rw, :w] (default: :r)
	:as — Determines which encoding/decoding functions to use when writing/reading values.
See explanation of automatic encoding/decoding.

Adds a read-only input register to a schema.
Opts
	:as — Determines which decoding functions to use when reading values.
See explanation of automatic encoding/decoding.

